

Historic, Archive Document

Do not assume content reflects current scientific knowledge, policies, or practices.

2521
.R44A7



United States
Department of
Agriculture

Agricultural
Research
Service

ARS-27

April 1985

A Fortran Program To Calculate Several Diversity Indices

CURRENT SERIAL RECORDS

MAR 20 '87

USDA
LIBRARY

ABSTRACT

Brown, M. W. 1985. A Fortran Program to Calculate Several Diversity Indices. U.S. Department of Agriculture, Agricultural Research Service, ARS-27, 24 p.

An interactive fortran program that will calculate Berger-Parker dominance index, species richness, Simpson index, Probability of Interspecific Encounter, Shannon-Weaver index, and Williams index is presented. The program also calculates the fit to the log-series distribution. The concept of diversity and the uses and limitations of each index are discussed. A sample run of the program is included with comments on the input and output.

Keywords: diversity, Berger-Parker dominance index, species richness, Simpson index, Probability of Interspecific Encounter, Shannon-Weaver index, Williams index, log-series

Copies of this publication can be purchased from the National Technical Information Service, 5285 Port Royal Road, Springfield, Virginia 22161.

ARS has no additional copies for free distribution.

A FORTRAN PROGRAM TO CALCULATE SEVERAL DIVERSITY INDICES¹

By M. W. Brown²

Diversity is a property of a collection of individuals that can be classified into mutually exclusive groups. Much of the theory and many indices were initially developed for use in information theory (Yule 1944; Shannon and Weaver 1949). Ecologists have adapted this information theory concept of diversity to investigate and describe the structure and function of communities in which individuals can be classified into species or other taxonomic and/or ecological groupings. Some of the same indices developed for information theory have been used in community ecology, but many other indices have been developed as well and the theoretical basis has been broadened (Fisher et al. 1943; Kempton and Taylor 1974; May 1975; Pielou 1977).

Diversity is generally considered to be composed of two elements: number of species or categories, and the relative abundances of those species (evenness). Herein lies the difficulty of developing a universal index of diversity; that is, incorporating both aspects into one number. Poole (1974) stated that the only objective measure of diversity is the number of species (species richness); once evenness is added, the index becomes dependent upon how evenness is defined. Another hazard in combining evenness and the number of species in a single index is that their individual effects may be confounded (King and Pimm 1983).

Despite these difficulties, diversity indices have a valuable role in community ecology. Because of different assumptions and definitions of evenness, each index provides a slightly different perspective of diversity. One must keep in mind the drawbacks of each index and select the index that best suits the task for which it is to be used. The researcher must first decide what he or she wants to know about the community and why (for example, to compare with other communities, to compare the same community over time, to evaluate the effect of a perturbation, or to analyze the structure or functioning of a particular community), and then select the most appropriate index. Southwood (1978) proposed a possible approach to selecting the proper index. For a general review of diversity and several of the indices, see Peet (1974), Poole (1974), Pielou (1977), or Southwood (1978).

¹This material is based upon work supported by the U.S. Department of Agriculture under Agreement No. 58-32U4-4-675.

²Adjunct Assistant Professor, West Virginia University, Appalachian Fruit Research Station, Route 2, Box 45, Kearneysville, WV 25430

The program presented here calculates a number of nonparametric diversity indices and one parametric index, Williams index. The nonparametric indices are not dependent upon the underlying distribution of the data, whereas Williams index is based on the log-series distribution. Other proposed distributions, such as the geometric (niche preemption model) (Poole 1974), broken stick model (MacArthur 1957), and log-normal (Preston 1948), are not used because the log-series seems to be more widely applicable and has a broader theoretical base (Taylor et al. 1976). The nonparametric indices calculated are species richness, Berger-Parker dominance index, Simpson index, Probability of Interspecific Encounter and Shannon-Weaver index.

The equations used assume the data are from a completely random sample of the community, containing all species which occur in the community. These are quite rigid assumptions, which unfortunately are seldom met. Although violation of the assumptions will bias the indices, they can still be useful as relative estimates of diversity if the limitations are kept in mind.

First, I present a discussion about each index, equations used, and references to each. The Fortran listing (from a Wang VS) is then given with an example and explanations of the input and output.

SPECIES RICHNESS

This index is the number of species (S) in the sample. It is only one part of what constitutes diversity, omitting evenness of abundances. However, it is a completely objective index as long as the sample is random and contains all species in the community.

BERGER-PARKER DOMINANCE INDEX (BERGER and PARKER 1970)

This index is only a measure of evenness with no information on the number of species present. The Berger-Parker index is defined as the probability that an individual selected at random from the community belongs to the most abundant species or, equivalently, the proportion of all individuals that are of the most abundant species. It is calculated as the largest P_i where $P_i = n_i/N$, N =the total number of individuals in the sample and n_i =the number of individuals in the sample belonging to species i . The index, which is easy to calculate and objective, is free of sample bias (Taylor et al. 1976), and characterizes the distribution of species abundances as well or better than other indices (May 1975).

SIMPSON INDEX
(SIMPSON 1949)

The Simpson index (C) is the probability that if two individuals are drawn at random from a community they will be of the same species. It is calculated as:

$$\Sigma [n_i (n_i - 1) / N(N-1)]$$

and has a variance of:

$$\frac{4N(N-1)(N-2)\Sigma p_i^3 + 2N(N-1)\Sigma p_i^2 - 2N(N-1)(2N-3)(\Sigma p_i^2)^2}{[N(N-1)]^2}$$

This index, like the Berger-Parker, is a measure of concentration or dominance. It is most influenced by the relative degree of abundance of the few most abundant (dominant) species rather than by the overall evenness of all species. Because it measures dominance, as diversity increases, C decreases. To transform this to an index (D) that is proportional to diversity, three conversion equations have been proposed. Pielou (1977) suggested $D = -\ln(C)$ in which form it is comparable to the Shannon-Weaver index and is a special case of a general diversity function given by Renyi (1961). Berger and Parker (1970) and Hurlbert (1971) suggested $D = 1 - C$ where D becomes the probability of the next encounter (the next individual sampled) being a different species. Finally, Southwood (1978) suggested $D = 1/C$. All three forms of D are calculated by the program.

Species evenness, defined as the ratio of observed diversity to maximum diversity (that is, all species in the sample being equally abundant), can be a useful measure. Evenness can range from a value of 0.0 if the observed diversity is at a minimum (that is, one species with all the individuals except for one individual in each of the other species) to 1.0 if the observed diversity is the maximum possible for the sample. Through the calculation of evenness, the diversity index is standardized, allowing for more valid comparisons among communities or among samples from the same community. Evenness, based on the Simpson index is calculated as:

$$C_{\max} = (S-1)/S$$

$$C_{\min} = (2N-S)(S-1)/N^2$$

$$\text{Evenness} = (C_{\max} - C_{\text{observed}}) / (C_{\max} - C_{\min})$$

PROBABILITY of
INTERSPECIFIC
ENCOUNTER (PIE)
(HURLBERT 1971)

This index is very similar to the Simpson index. PIE is defined as the proportion of all the possible encounters among individuals that are interspecific. The difference between the Simpson index and PIE is the same as sampling from a finite population with replacement (Simpson index) or without replacement (PIE). For all but very small data sets $PIE = 1 - \text{Simpson index}$, but they are different conceptually. PIE is calculated as:

$$\sum (n_i/N) [(N-n_i)/(N-1)]$$

The variance is calculated as the variance of the Simpson index multiplied by $[N/(N-1)]^2$. Species evenness can be defined as $(PIE - PIE_{\min}) / (PIE_{\max} - PIE_{\min})$ with the following formulas:

$$PIE_{\max} = [N/(N-1)] [(S-1)/S]$$

$$PIE_{\min} = [N/(N-1)] [(2N-S)(S-1)/N^2]$$

SHANNON-WEAVER
INDEX (SHANNON
AND WEAVER 1949)

This index is the most commonly used and accepted in ecological work. It has been criticized, however, as having no intuitive ecological meaning (Taylor et al. 1976). Southwood (1978) considered it to be a distraction rather than an asset in ecological use. It is easily calculated, and it is recognized by most, if not all, ecologists. The Shannon-Weaver index (H') is a measure of uncertainty in predicting the species of an individual drawn at random from the community. Two equations are used by the program. The first:

$$-\sum p_i [\ln(p_i)]$$

is the one most used but is biased. The second is the expected value of H' . Only the first four terms of the expansion given by Hutcheson (1970) are used in the program.

$$-\sum p_i [\ln(p_i)] - (S-1)/2N + (1 - \sum p_i^{-1})/12N^2 + \sum (p_i^{-1} - p_i^{-2})/12N^3$$

The variance of $E(H')$, using the first three terms of Hutcheson's (1970) expansion, is:

$$(\sum p_i [\ln(p_i)]^2 - \{\sum p_i [\ln(p_i)]\}^2)/N + (S-1)/2N^2 +$$

$$(-1 + \sum p_i^{-1} - \{\sum p_i^{-1} [\ln(p_i)]\} + \sum p_i^{-1} \{\sum p_i [\ln(p_i)]\})/6N^3$$

The maximum diversity for a given community is $H'_{\max} = \ln(S)$. This can be used to estimate evenness of the community, H'/H'_{\max} . Strictly speaking, the index and its variance are inestimable when the true number of species in the community is unknown.

WILLIAMS INDEX
(FISHER ET AL.
1943)

Williams index is the only parametric diversity index calculated by this program, the index being a parameter (α) of the log-series distribution. A justification for the use of this index is given by Kempton and Taylor (1974) and Taylor et al. (1976). They stated that this index should be the index of choice, assuming a log-series distribution a priori. The log-series, an approximation to the Gamma distribution, has two parameters: α (Williams index), and X . Alpha is a characteristic of the community. The value of α is most influenced by the moderately abundant species. Therefore, it is relatively insensitive to rare or dominant species, the presence and abundance of which are more subject to sampling errors. X is a characteristic of the sample itself and is dependent on the total number of individuals.

Alpha is calculated by iteration of the maximum likelihood estimator (Taylor et al. 1976):

$$S = \alpha[\ln(1+N/\alpha)]$$

X is calculated by solving equations (Pielou 1977):

$$S = -\alpha[\ln(1-X)]$$

and

$$N = \alpha X/(1-X)$$

for α , and iterating the result:

$$S = \{[N(1-X)]/X\} [-\ln(1-X)]$$

The program prints out the predicted number of species from each iteration. These values can be used to check for errors if the predicted S is very different from the actual S . Another check can be made by using figure 8 of Fisher et al. (1943). Variance of α is calculated by

$$-\alpha/\ln(1-X)$$

Calculation of the expected log-series distribution using the calculated parameters α and X is by

$$E(r) = \alpha X^r / r$$

where $E(r)$ is the expected number of species with r individuals. A chi-square goodness of fit test is conducted on the data, combined into octaves (see Poole 1974), to judge whether α should be used as the diversity index. Finally, expected values of S and N based on the log-series distribution are calculated as:

$$S = -\alpha [\ln(1-X)]$$

$$N = \alpha X / (1-X)$$

LITERATURE CITED

- Berger, W.H., and F.L. Parker. 1970. Diversity of planktonic Foraminifera in deep-sea sediments. *Science* 168:1345-1347.
- Fisher, R.A., A.S. Corbet, and C.B. Williams. 1943. The relation between the number of species and the number of individuals in a random sample of an animal population. *J. Anim. Ecol.* 12:42-58.
- Hurlbert, S.H. 1971. The nonconcept of species diversity: a critique and alternative parameters. *Ecology* 52:577-586.
- Hutcheson, K. 1970. A test for comparing diversities based on the Shannon formula. *J. Theor. Biol.* 29:151-154.
- Kempton, R.A., and L.R. Taylor. 1974. Log-series and log-normal parameters as diversity discriminants for the Lepidoptera. *J. Anim. Ecol.* 43:381-399.
- King, A.W., and S.L. Pimm. 1983. Complexity, diversity, and stability: a reconciliation of theoretical and empirical results. *Am. Nat.* 122:229-239.
- MacArthur, R.H. 1957. On the relative abundance of bird species. *Proc. Nat. Acad. Sci. Wash.* 43:293-295.
- May, R.M. 1975. Patterns of species abundance and diversity. In: Cody, M.L. and J.M. Diamond (eds.), *Ecology and Evolution of Communities*. Harvard Univ. Press, Cambridge, Mass.
- Peet, R.K. 1974. The measurement of species diversity. *Ann. Rev. Ecol. Syst.* 5:285-307.
- Pielou, E.C. 1977. *Mathematical Ecology*. John Wiley & Sons, New York.
- Poole, R.W. 1974. *An Introduction to Quantitative Ecology*. McGraw-Hill, New York.
- Preston, F.W. 1948. The commonness and rarity of species. *Ecology* 29:254-283.
- Renyi, A. 1961. On measures of entropy and information. *Proc. 4th. Berkeley Symposium on Math. Stat. and Prob.* 1:547-561. (as cited by Pielou 1977).
- Shannon C.E., and W. Weaver. 1949. *The Mathematical Theory of Communication*. Univ. of Illinois Press, Urbana, Ill.

Simpson, E.H. 1949. Measurement of diversity. *Nature* 163:688.

Southwood, T.R.E. 1978. *Ecological Methods with Particular Reference to the Study of Insect Populations*. 2nd edition. Chapman and Hall, New York.

Taylor, L.R., R.A. Kempton, and I.P. Woiwod. 1976. Diversity statistics and the log-series model. *J. Anim. Ecol.* 45:255-272.

Yule, G.U. 1944. *The Statistical Study of Vocabulary*. Cambridge University Press, London.

APPENDIX A: PROGRAM INPUT

Note: Computer generated comments and questions are in capital letters; user responses are in lowercased letters or numerals. Comments by the author are offset with an asterisk.

INPUT TITLE (UP TO 32 CHARACTERS)

abandoned orchard wva, 18-iv-83

*Data on relative abundance of herbivorous insects found in an abandoned apple orchard in West Virginia collected by M. W. Brown and Cynthia R. Loerch, Appalachian Fruit Research Station, Kearneysville, WV.

INPUT NUMBER OF QUADRATS. (TWO DIGITS)

06

*Zeros are put in to show spacing for the proper format and are not needed when supplying data.

INPUT TOTAL NUMBER OF SPECIES. (TWO DIGITS)

12

*This is the number of species found on all quadrats in the sample.

BEGIN DATA ENTRY.

INPUT DATA FOR SPECIES 1

000700010002000000040003

*Zeros are included only to show spacing. There are 6 fields (number of fields taken from the number of quadrats supplied earlier) of 4 digits (F4.0 format).

INPUT DATA FOR SPECIES 2

000300030006000600080007

INPUT DATA FOR SPECIES 3

000400050005000700020000

INPUT DATA FOR SPECIES 4

001000000001000000000000

INPUT DATA FOR SPECIES 5

000000000001000000000000

INPUT DATA FOR SPECIES 6

000000000003000200030004

INPUT DATA FOR SPECIES 7

0000000000000000100010000

INPUT DATA FOR SPECIES 8

0000000000000000100000000

INPUT DATA FOR SPECIES 9

0000000000000000100000000

INPUT DATA FOR SPECIES 10

000000000000000000030005

INPUT DATA FOR SPECIES 11

0000000000000000000000001

INPUT DATA FOR SPECIES 12

0000000000000000000000001

DO YOU WANT TO CHECK DATA? Y/N

y

SPECIES

QUADRAT

	1	2	3	4	5	6
1	7.	1.	2.	0.	4.	3.
2	3.	3.	6.	6.	8.	7.
3	4.	5.	5.	7.	2.	0.
4	10.	0.	1.	0.	0.	0.
5	0.	0.	1.	0.	0.	0.

6	0.	0.	3.	2.	3.	4.
7	0.	0.	0.	1.	1.	0.
8	0.	0.	0.	1.	0.	0.
9	0.	0.	0.	1.	0.	0.
10	0.	0.	0.	0.	3.	5.
11	0.	0.	0.	0.	0.	1.
12	0.	0.	0.	0.	0.	1.

DO YOU WANT TO CORRECT ANY DATA? Y/N

y

WHICH QUADRAT? (TWO DIGITS)

01

WHICH SPECIES? (TWO DIGITS)

04

ENTER DATA FOR QUADRAT 1 SPECIES 4

0000

DO YOU WANT TO CORRECT ANY OTHER DATA? Y/N

n

DO YOU WANT TO LOOK AT THE DATA AGAIN? Y/N

n

*At this point you could examine the data again and correct more errors by answering "Y".

DIVERSITY CALCULATIONS IN PROGRESS

APPENDIX B: PROGRAM OUTPUT

ABANDONED ORCHARD WVA, 18-IV-83

***** DATA *****

SPECIES

QUADRAT

1 2 3 4 5 6

1	7.	1.	2.	0.	4.	3.
2	3.	3.	6.	6.	8.	7.
3	4.	5.	5.	7.	2.	0.
4	0.	0.	1.	0.	0.	0.
5	0.	0.	1.	0.	0.	0.
6	0.	0.	3.	2.	3.	4.
7	0.	0.	0.	1.	1.	0.
8	0.	0.	0.	1.	0.	0.
9	0.	0.	0.	1.	0.	0.
10	0.	0.	0.	0.	3.	5.
11	0.	0.	0.	0.	0.	1.
12	0.	0.	0.	0.	0.	1.

***** DIVERSITY STATISTICS *****

SPECIES	NUMBER	PROBABILITY
1	17.	0.1683
2	33.	0.3267
3	23.	0.2277
4	1.	0.0099
5	1.	0.0099
6	12.	0.1188
7	2.	0.0198
8	1.	0.0099
9	1.	0.0099
10	8.	0.0792
11	1.	0.0099
12	1.	0.0099

SPECIES RICHNESS = 12 101. INDIVIDUALS

BERGER-PARKER DOMINANCE INDEX = 0.3267

SIMPSON INDEX OF DIVERSITY = 0.2004 VARIANCE = 0.000435

1/SIMPSON INDEX = 4.9901
1-SIMPSON INDEX = 0.7996
-LN SIMPSON INDEX = 1.6075
MINIMUM SIMPSON = 0.0742
MAXIMUM SIMPSON = 0.7931
EVENNESS = 0.8244

PROBABILITY OF INTERSPECIFIC ENCOUNTER = 0.7996 VARIANCE = 0.000443

MAXIMUM P.I.E. = 0.9258
MINIMUM P.I.E. = 0.2069
EVENNESS = 0.8244

SHANNON-WEAVER INDEX H= 1.8081

EXPECTED SHANNON-WEAVER INDEX = 1.7429 VARIANCE = 0.008775
MAXIMUM DIVERSITY = 2.4849
EVENNESS = 0.7276

***** LOG-SERIES STATISTICS *****

ALPHA (WILLIAMS INDEX) = 3.55 (PREDICTED NUMBER OF SPECIES = 12.0086)
 X = 0.9661 (PREDICTED NUMBER OF SPECIES = 11.9943)
 VARIANCE OF ALPHA = 1.048949

CHI-SQUARE GOODNESS OF FIT TABLE

INDIVIDUALS PER SPECIES (OCTAVE)	NUMBER OF SPECIES EXPECTED	SPECIES OBSERVED	CHI-SQUARE
1	3.430	6	1.9263
2-3	2.724	1	1.0909
4-7	2.250	0	2.2501
8-15	1.766	2	0.0311
16-31	1.170	2	0.5885
32-63	0.541	1	0.3890
64+	0.125	0	0.1255

GOODNESS OF FIT TO THE LOG-SERIES MODEL CHI-SQUARE = 6.4014

EXPECTED NUMBER OF SPECIES = 12.0144
 EXPECTED NUMBER OF INDIVIDUALS = 101.1695

APPENDIX C: PROGRAM LISTING

```

C JOB CONTROL STATEMENTS MUST BE PLACED HERE
C
C PROGRAM TO READ DATA ON SPECIES ABUNDANCE PER QUADRAT AND
C CALCULATE DIVERSITY INDICES AND FIT TO THE LOG-SERIES MODEL.
C
C PROGRAM VARIABLES ARE DIMENSIONED TO HANDLE UP TO 100 SPECIES AND
C 10 QUADRATS BUT IS FORMATTED FOR 10 SPECIES WITH AN ABUNDANCE OF
C UP TO 9999. INDIVIDUALS. THESE RESTRICTIONS ARE ONLY FOR FORMATTING
C AND MAY BE ALTERED BY EDITING THE CORRESPONDING DIMENSION AND
C FORMAT STATEMENTS.
C
      INTEGER Q(10),OBS(7)
      REAL N(100,10),SPLIST(100),P(100),EXPECT(7),CHI(7)
      REAL*8 A,B,C,D,E
      DATA N/1000*0.0/,BERG/0.0/,SPLIST/100*0.0/,TOTAL/0.0/,RRRP/0.0/
      DATA OBS/7*0/,SIMP/0.0/,SHAN/0.0/,RP/0.0/,PSQ/0.0/,RRP/0.0/
      DATA PIE/0.0/,SUMP3/0.0/,SUMP2/0.0/,VARSIM/0.0/
      WRITE(0,900)
900  FORMAT(' INPUT TITLE      (UP TO 32 CHARACTERS)')
      READ(0,902) A,B,C,D
902  FORMAT(4A8)
      WRITE(6,904) A,B,C,D
904  FORMAT('1',5X,4A8)
      WRITE(0,906)
906  FORMAT(' INPUT NUMBER OF QUADRATS. (TWO DIGITS)')
      READ(0,908) J
908  FORMAT(I2)
      WRITE(0,910)
910  FORMAT(' INPUT TOTAL NUMBER OF SPECIES. (TWO DIGITS)')
      READ(0,912) I
912  FORMAT(I2)
      S=I
      WRITE(0,914)
914  FORMAT(' BEGIN DATA ENTRY.')
      DO 100 JJ=1,J
        Q(JJ)=JJ
100  CONTINUE
      DO 102 II=1,I
        WRITE(0,916) II
916  FORMAT(' INPUT DATA FOR SPECIES',I3)
        READ(0,918) (N(II,JJ),JJ=1,J)
918  FORMAT(10F4.0)
102  CONTINUE
      WRITE(0,920)
920  FORMAT(' DO YOU WANT TO CHECK DATA?  Y/N')
      READ(0,922) E
922  FORMAT(A8)
      IF(E.NE.'Y') GO TO 304
300  WRITE(0,924)
924  FORMAT(////,' SPECIES',10X,'QUADRAT')
      WRITE(0,926) (Q(JJ),JJ=1,J)
926  FORMAT(//,9X,10(I2,3X),//)

```

```

DO 104 II=1,I
WRITE(0,928) II,(N(II,JJ),JJ=1,J)
928 FORMAT(I5,2X,10F5.0)
IF (II.EQ.20) PAUSE
IF (II.EQ.40) PAUSE
IF (II.EQ.60) PAUSE
104 CONTINUE
WRITE(0,930)
930 FORMAT(' DO YOU WANT TO CORRECT ANY DATA? Y/N')
READ(0,922) E
IF(E.NE.'Y') GO TO 304
302 WRITE(0,932)
932 FORMAT(' WHICH QUADRAT? (TWO DIGITS)')
READ(0,934) JJ
934 FORMAT(I2)
WRITE(0,936)
936 FORMAT(' WHICH SPECIES? (TWO DIGITS)')
READ(0,938) II
938 FORMAT(I2)
WRITE(0,940) JJ,II
940 FORMAT(' ENTER DATA FOR QUADRAT',I2,' SPECIES',I3)
READ(0,942) N(II,JJ)
942 FORMAT(F4.0)
WRITE(0,944)
944 FORMAT(' DO YOU WANT TO CORRECT ANY OTHER DATA? Y/N')
READ(0,922) E
IF(E.EQ.'Y') GO TO 302
WRITE(0,946)
946 FORMAT(' DO YOU WANT TO LOOK AT THE DATA AGAIN? Y/N')
READ(0,922) E
IF(E.EQ.'Y') GO TO 300
304 WRITE(6,948)
948 FORMAT(//////,25X,'***** DATA *****',///,' SPECIES',
120X,'QUADRAT')
WRITE(6,950) (Q(JJJ),JJJ=1,J)
950 FORMAT(//,10X,18I7)
WRITE(6,952)
952 FORMAT(/)
DO 106 III=1,I
WRITE(6,954) III,(N(III,JJJ),JJJ=1,J)
954 FORMAT(I5,6X,18F7.0)
106 CONTINUE
WRITE(0,956)
956 FORMAT('1',' DIVERSITY CALCULATIONS IN PROGRESS')
DO 108 K=1,I
DO 110 KK=1,J
SPLIST(K)=SPLIST(K)+N(K,KK)
110 CONTINUE
108 CONTINUE
DO 112 K=1,I
TOTAL=TOTAL+SPLIST(K)
112 CONTINUE
DO 114 K=1,I
P(K)=SPLIST(K)/TOTAL
SUMP2=SUMP2+P(K)*P(K)
SUMP3=SUMP3+P(K)**3.
RP=RP+(1./P(K))

```

```

RRP=RRP+((1./P(K))-(1./(P(K)*P(K))))
SIMP=SIMP+(SPLIST(K)*(SPLIST(K)-1.)/(TOTAL*(TOTAL-1.))
PIE=PIE+(SPLIST(K)/TOTAL)*((TOTAL-SPLIST(K))/(TOTAL-1.))
IF(P(K).GT.BERG) BERG=P(K)
IF(SPLIST(K).LE.0.0) GO TO 114
SHAN=SHAN-(P(K)*ALOG(P(K)))
PSQ=PSQ+(P(K)*ALOG(P(K))*ALOG(P(K)))
RRRP=RRRP+((1./P(K))*ALOG(P(K)))
114 CONTINUE
RSIMP1=1./SIMP
RSIMP2=1.-SIMP
RSIMP3=-(ALOG(SIMP))
VARSIM=((4.*TOTAL*(TOTAL-1.)*(TOTAL-2.)*SUMP3)+(2.*TOTAL*(TOTAL-1.
1)*SUMP2)-(2.*TOTAL*(TOTAL-1.)*((2.*TOTAL)-3.)*(SUMP2*SUMP2)))/
2((TOTAL*(TOTAL-1.))*(TOTAL*(TOTAL-1.)))
VARPIE=(TOTAL/(TOTAL-1.))*(TOTAL/(TOTAL-1.))*VARSIM
PIEMAX=(TOTAL/(TOTAL-1.))*((S-1.)/S)
PIEMIN=(TOTAL/(TOTAL-1.))*(((2.*TOTAL)-S)*(S-1.))/(TOTAL*TOTAL)
PIEEVN=(PIE-PIEMIN)/(PIEMAX-PIEMIN)
SIMMAX=((TOTAL-S+1.)*(TOTAL-S))/(TOTAL*(TOTAL-1.))
SIMMIN=S*(((TOTAL/S)*((TOTAL/S)-1.))/(TOTAL*(TOTAL-1.)))
SIMEVN=(SIMMAX-SIMP)/(SIMMAX-SIMMIN)
SHANSQ=SHAN*SHAN
ESHAN=SHAN-((I-1.)/(2*TOTAL))+((1.-RP)/(12.*TOTAL*TOTAL))+
1(RRP/(12.*TOTAL**3.))
VARSHA=((PSQ-SHANSQ)/TOTAL)+((I-1.)/(2.*TOTAL*TOTAL))+
1(-1.+RP-RRRP-(RP*SHAN))/(6.*TOTAL**3.)
HMAX=ALOG(S)
EVEN=SHAN/HMAX
WRITE(6,958)
958 FORMAT('1',20X,'***** DIVERSITY STATISTICS *****',///,
1' SPECIES          NUMBER          PROBABILITY')
DO 116 K=1,I
WRITE(6,960) K,SPLIST(K),P(K)
960 FORMAT(' ',I7,10X,F6.0,10X,F10.4)
116 CONTINUE
WRITE(6,962) I,TOTAL
962 FORMAT(///,' SPECIES RICHNESS = ',I5,F10.0,' INDIVIDUALS')
WRITE(6,964) BERG
964 FORMAT(///,' BERGER-PARKER DOMINANCE INDEX = ',F10.4)
WRITE(6,966) SIMP,VARSIM,RSIMP1,RSIMP2,RSIMP3,SIMMIN,SIMMAX,SIMEVN
966 FORMAT(///,' SIMPSON INDEX OF DIVERSITY = ',F15.4,5X,' VARIANCE = ',
1F10.6,/,6X,' 1/SIMPSON INDEX = ',F9.4,/,6X,' 1-SIMPSON INDEX = ',
2F9.4,/,6X,' -LN SIMPSON INDEX = ',F7.4,/,6X,' MINIMUM SIMPSON = ',
3F9.4,/,6X,' MAXIMUM SIMPSON = ',F9.4,/,6X,' EVENNESS = ',F16.4)
WRITE(6,968) PIE,VARPIE,PIEMAX,PIEMIN,PIEEVN
968 FORMAT(///,' PROBABILITY OF INTERSPECIFIC ENCOUNTER = ',F15.4,
15X,' VARIANCE = ',F10.6,/,6X,' MAXIMUM P.I.E. = ',F12.4,/,6X,
2' MINIMUM P.I.E. = ',F12.4,/,6X,' EVENNESS = ',F18.4)
WRITE(6,970) SHAN,ESHAN,VARSHA,HMAX,EVEN
970 FORMAT(///,' SHANNON-WEAVER INDEX H= ',F10.4,/,5X,' EXPECTED
1 SHANNON-WEAVER INDEX = ',F10.4,5X,' VARIANCE = ',F10.6,/,5X,
2' MAXIMUM DIVERSITY = ',F9.4,/,5X,' EVENNESS = ',F10.4)
C SECTION TO ESTIMATE ALPHA OF THE LOG-SERIES DISTRIBUTION
C USES MAXIMUM LIKELIHOOD ESTIMATOR
C S=ALPHA*(LOGE(1+N/ALPHA))
D1=99999999.

```

```

DO 118 K=1,10000
ALPHA=0.1+K/100.
T=ALPHA*ALOG(1.+TOTAL/ALPHA)
D2=ABS(S-T)
IF (D2.GE.D1) GO TO 306
D1=D2
XT=T
XALPHA=ALPHA
118 CONTINUE
306 WRITE(6,972)
972 FORMAT('1',20X,'***** LOG-SERIES STATISTICS *****')
WRITE(6,974) XALPHA,XT
974 FORMAT(///,' ALPHA (WILLIAMS INDEX) = ',F11.2,' (PREDICTED NUMBER
1 OF SPECIES = ',F10.4,' )')
C SECTION TO ESTIMATE X OF THE LOG-SERIES DISTRIBUTION
C USES MAXIMUM LIKELIHOOD ESTIMATOR
C S=(N(1-X)/X)*(-LOGE(1-X))
T=0.0
D1=999999999.
X=0.1
DO 120 K=1,8999
X=0.1+K/10000.
T=-((TOTAL*(1.-X))/X)*(ALOG(1.-X))
D2=ABS(S-T)
IF (D2.GE.D1) GO TO 308
D1=D2
XT=T
XX=X
120 CONTINUE
308 WRITE(6,976) XX,XT
976 FORMAT(/,' X',21X,'=',F12.4,' (PREDICTED NUMBER OF SPECIES = ',
1F10.4,' )')
VARA=-XALPHA/ALOG(1.-XX)
WRITE(6,978) VARA
978 FORMAT(/,' VARIANCE OF ALPHA =',F12.6)
C SECTION TO CALCULATE FIT TO THE LOG-SERIES MODEL
EXPECT(1)=XALPHA*XX
EXPECT(2)=(XALPHA*(XX**2.))/2.+(XALPHA*(XX**3.))/3.
EXPECT(3)=0.
DO 122 K=1,4
EXPECT(3)=EXPECT(3)+(XALPHA*(XX**(K+3.)))/(K+3.)
122 CONTINUE
EXPECT(4)=0.
DO 124 K=1,8
EXPECT(4)=EXPECT(4)+(XALPHA*(XX**(K+7.)))/(K+7.)
124 CONTINUE
EXPECT(5)=0.
DO 126 K=1,16
EXPECT(5)=EXPECT(5)+(XALPHA*(XX**(K+15.)))/(K+15.)
126 CONTINUE
EXPECT(6)=0.
DO 128 K=1,32
EXPECT(6)=EXPECT(6)+(XALPHA*(XX**(K+31.)))/(K+31.)
128 CONTINUE
EXPECT(7)=0.
DO 130 K=1,64
EXPECT(7)=EXPECT(7)+(XALPHA*(XX**(K+63.)))/(K+63.)

```

```

130 CONTINUE
C SECTION TO CALCULATE FIT TO THE LOG-SERIES MODEL
DO 132 M=1,I
  IF (SPLIST(M).EQ.1.) OBS(1)=OBS(1)+1
  IF (SPLIST(M).EQ.2..OR.SPLIST(M).EQ.3.) OBS(2)=OBS(2)+1
  IF (SPLIST(M).GE.4..AND.SPLIST(M).LT.8.) OBS(3)=OBS(3)+1
  IF (SPLIST(M).GE.8..AND.SPLIST(M).LT.16.) OBS(4)=OBS(4)+1
  IF (SPLIST(M).GE.16..AND.SPLIST(M).LT.32.) OBS(5)=OBS(5)+1
  IF (SPLIST(M).GE.32..AND.SPLIST(M).LT.64.) OBS(6)=OBS(6)+1
  IF (SPLIST(M).GE.64.) OBS(7)=OBS(7)+1
132 CONTINUE
CHISQ=0.0
WRITE(6,980)
980 FORMAT(////,20X,'CHI-SQUARE GOODNESS OF FIT TABLE',///,
1' INDIVIDUALS PER SPECIES          NUMBER OF SPECIES          CHI-
2SQUARE',/,7X,'(OCTAVE)',16X,'EXPECTED          OBSERVED',/)
DO 134 M=1,7
  CHI(M)=(OBS(M)-EXPECT(M))*(OBS(M)-EXPECT(M))/EXPECT(M)
  CHISQ=CHISQ+CHI(M)
134 CONTINUE
WRITE(6,982) (EXPECT(M),OBS(M),CHI(M),M=1,7)
982 FORMAT(' ',8X,'1',19X,F10.3,I10,F20.4,/,8X,'2-3',18X,F10.3,I10,
1F20.4,/,8X,'4-7',18X,F10.3,I10,F20.4,/,8X,'8-15',17X,F10.3,
2I10,F20.4,/,7X,'16-31',17X,F10.3,I10,F20.4,/,7X,'32-63',
317X,F10.3,I10,F20.4,/,7X,'64+',19X,F10.3,I10,F20.4)
WRITE(6,984) CHISQ
984 FORMAT(//,' GOODNESS OF FIT TO THE LOG-SERIES MODEL',
1' CHI-SQUARE =',F12.4)
EXPS=-XALPHA*ALOG(1.-XX)
EXPN=XALPHA*XX/(1.-XX)
WRITE(6,986) EXPS,EXPN
986 FORMAT(//,' EXPECTED NUMBER OF SPECIES = ',F19.4,/,
1' EXPECTED NUMBER OF INDIVIDUALS = ',F14.4)
C
C FOR SOME COMPILERS A "STOP" STATEMENT MAY BE NEEDED HERE.
C
END

```


EXCHANGE Rec'd

14 1987